

Actividad Guiada: Sistema experto básico con Prolog

Antes de empezar...

Revisa y realiza la actividad [UT3 A1 Introducción a la programación lógica. Prolog](#).

Revisa la documentación de Prolog y sistemas expertos proporcionada como recurso en el Moodle.

Qué vamos a hacer

Vamos a programar un sistema experto básico para el diagnóstico de enfermedades a partir de los síntomas observados. Nuestro sistema no tendrá ninguna regla avanzada ni ningún tipo de tratamiento de incertidumbre, y su función básica será comprobar el funcionamiento de la interfaz de usuario.

1. Creación de la base de conocimiento

Vamos a crear primero la base de conocimiento de nuestro sistema. En ella incluiremos los síntomas que determinan el diagnóstico para cada una de las enfermedades que tratamos en el sistema experto.

```
/* Base de Conocimiento: enfermedades.pl
Diagnosticos y sintomas
Dominio: diagnostico medico. Trata los sintomas como una lista multi-elementos
*/

conocimiento('sarampion',
['el paciente esta cubierto de puntos', 'el paciente tiene fiebre', 'el
paciente tiene ojos rojos', 'el paciente tiene tos seca']).

conocimiento('catarro',
['el paciente tiene secrecion nasal', 'el paciente tiene tos seca', 'el
paciente tiene dolor de cabeza', 'el paciente tiene estornudos']).

conocimiento('malaria',
['el paciente tiene fiebre', 'el paciente tiene dolor en las articulaciones',
'el paciente tiembla violentamente', 'el paciente tiene escalofrios']).

conocimiento('gripe',
['el paciente tiene dolor en las articulaciones', 'el paciente tiene dolor de
cabeza', 'el paciente tiene tos seca', 'el paciente tiene fiebre', 'el
paciente tiene escalofrios']).

conocimiento('tifoidea',
['el paciente tiene falta de apetito', 'el paciente tiene fiebre', 'el
paciente tiene dolor abdominal', 'el paciente tiene dolor de cabeza', 'el
paciente tiene diarrea']).
```

Mediante una serie de preguntas podremos incluir los síntomas que tendrá el paciente y nuestro sistema podrá ayudarnos a identificar (o no) la enfermedad que padece.

Mirando el contenido más detenidamente, podemos ver cómo cada una de las enfermedades se identifican como una lista de síntomas. Nuestro conocimiento sobre, por ejemplo, el

sarampión incluye sus síntomas (incluidos en la lista ['el paciente esta cubierto de puntos', 'el paciente tiene temperatura alta', 'el paciente tiene ojos rojos', 'el paciente tiene tos seca']). Creamos un archivo *enfermedades.pl* en el incluiremos el código anterior como parte de nuestra base de conocimiento.

2. Creación de la interfaz de usuario

Seguidamente tendremos que programar la interfaz de interacción con el usuario del sistema experto.

Crea un archivo *experto.pl* que contenga lo siguiente:

```
/* Sistema Experto: experto.pl
Trata los sintomas como una lista. La cabeza es el diagnostico y la "cola" son
los sintomas.
Utiliza assert/1 para cambiar dinamicamente la base de conocimientos.
Determina la verdad y falsedad de los sintomas conocidos.
Puede contestar a las preguntas 'porque' e incluye capacidad de explicacion.
Elimina dinamicamente las aseveraciones agregadas despues de cada consulta.
*/
:- dynamic conocido/1.

consulta:- haz_diagnostico(X), escribe_diagnostico(X),
ofrece_explicacion_diagnostico(X),
clean_scratchpad.

consulta:- write('No hay suficiente conocimiento para elaborar un
diagnostico.'), clean_scratchpad.

haz_diagnostico(Diagnosis):- obten_hipotesis_y_sintomas(Diagnosis,
ListaDeSintomas),
prueba_presencia_de(Diagnosis, ListaDeSintomas).

obten_hipotesis_y_sintomas(Diagnosis, ListaDeSintomas):-
conocimiento(Diagnosis, ListaDeSintomas).

prueba_presencia_de(Diagnosis, []).

prueba_presencia_de(Diagnosis, [Head | Tail]):- prueba_verdad_de(Diagnosis,
Head),
prueba_presencia_de(Diagnosis, Tail).

prueba_verdad_de(Diagnosis, Sintoma):- conocido(Sintoma).

prueba_verdad_de(Diagnosis, Sintoma):- not(conocido(is_false(Sintoma))),
pregunta_sobre(Diagnosis, Sintoma, Reply), Reply = si.

pregunta_sobre(Diagnosis, Sintoma, Reply):- write('Es verdad que '),
write(Sintoma), write('? '),
read(Respuesta), process(Diagnosis, Sintoma, Respuesta, Reply).

process(Diagnosis, Sintoma, si, si):- asserta(conocido(Sintoma)).

process(Diagnosis, Sintoma, no, no):- asserta(conocido(is_false(Sintoma))).

process(Diagnosis, Sintoma, porque, Reply):- nl,
write('Estoy investigando la hipotesis siguiente: '),
write(Diagnosis), write('.'), nl, write('Para esto necesito saber si '),
write(Sintoma), write('.'), nl, pregunta_sobre(Diagnosis, Sintoma, Reply).

process(Diagnosis, Sintoma, Respuesta, Reply):- Respuesta \== no,
Respuesta \== si, Respuesta \== porque, nl,
write('Debes contestar si, no o porque.'), nl,
pregunta_sobre(Diagnosis, Sintoma, Reply).
```

```

escribe_diagnostico(Diagnosis):- write('El diagnostico es '),
write(Diagnosis), write('.'), nl.
ofrece_explicacion_diagnostico(Diagnosis):-
pregunta_si_necesita_explicacion(Respuesta),
actua_consecuentemente(Diagnosis, Respuesta).
pregunta_si_necesita_explicacion(Respuesta):-
write('Quieres que justifique este diagnostico? '),
read(RespuestaUsuario),
asegura_respuesta_si_o_no(RespuestaUsuario, Respuesta).

asegura_respuesta_si_o_no(si, si).

asegura_respuesta_si_o_no(no, no).

asegura_respuesta_si_o_no(_, Respuesta):- write('Debes contestar si o no.'),
pregunta_si_necesita_explicacion(Respuesta).

actua_consecuentemente(Diagnosis, no).

actua_consecuentemente(Diagnosis, si):- conocimiento(Diagnosis,
ListaDeSintomas),
write('Se determino este diagnostico porque se encontraron los siguientes
sintomas: '), nl, escribe_lista_de_sintomas(ListaDeSintomas).

escribe_lista_de_sintomas([]).

escribe_lista_de_sintomas([Head | Tail]):-
write(Head), nl, escribe_lista_de_sintomas(Tail).

clean_scratchpad:- retract(conocido(X)), fail.

clean_scratchpad.

conocido(_):- fail.

not(X):- X,!,fail.

not(_).

```

Nota: si encuentras problemas en el código tras copiar/pegar el texto, revisa bien las comillas

Como hemos visto anteriormente, Prolog va desarrollando las cláusulas de adelante hacia atrás. Para hacer la interfaz, creamos primero una regla que será la que lance la interfaz de interacción con el usuario, en este caso 'consulta'. Cuando escribamos 'consulta.' en el intérprete se desarrollará lo que está al otro lado de la cláusula 'consulta' para comprobar si es verdad, que en este caso son otras reglas llamadas 'haz_diagnostico(X)', 'escribe_diagnostico(X)' y 'ofrece_explicacion_diagnostico(X)', para finalmente limpiar la pantalla, todo en ese orden. Cada una de esas reglas se irá desarrollando de manera recursiva ('haz_diagnostico(X)' se desarrollará primero con 'obten_hipotesis_y_sintomas(X, ListaDeSintomas)', y así sucesivamente).

3. Cargar los archivos y ejecutar el programa

Vamos ahora a cargar los archivos y a ejecutar la consulta que nos permita hacer el diagnóstico. Cargamos los dos archivos generados anteriormente con la base de conocimiento y el shell de usuario. Podemos hacerlo bien cargando mediante la interfaz gráfica (File->Consult...) o escribiendo *consult*("nombre_del_archivo").

Una vez cargados ambos archivos ejecutamos la consulta escribiendo:

? - consulta.

A partir de ahí el intérprete irá preguntándonos por los síntomas para intentar hacer un diagnóstico. **RECUERDA AGREGAR SIEMPRE EL PUNTO AL FINAL DE CADA RESPUESTA.**

```
?- consulta.
Es verdad que el paciente esta cubierto de puntos? no.
Es verdad que el paciente tiene secrecion nasal? |: no.
Es verdad que el paciente tiene fiebre? |: si.
Es verdad que el paciente tiene dolor en las articulaciones? |: porque.

Estoy investigando la hipotesis siguiente: malaria.
Para esto necesito saber si el paciente tiene dolor en las articulaciones.
Es verdad que el paciente tiene dolor en las articulaciones? |: no.
Es verdad que el paciente tiene falta de apetito? |:
```

Una vez llega a un diagnóstico exitoso, nos ofrecerá la posibilidad de explicar por qué ha sacado esa conclusión antes de finalizar.

```
Es verdad que el paciente tiene dolor abdominal? |: si.
Es verdad que el paciente tiene dolor de cabeza? |: si.
Es verdad que el paciente tiene diarrea? |: si.
El diagnostico es tifoidea.
Quieres que justifique este diagnostico? |: si.
Se determino este diagnostico porque se encontraron los siguientes
síntomas:
el paciente tiene falta de apetito
el paciente tiene fiebre
el paciente tiene dolor abdominal
el paciente tiene dolor de cabeza
el paciente tiene diarrea
true .
```

Prueba el sistema con los siguientes síntomas:

1. Secreción nasal, tos seca, dolor de cabeza, estornudos.
2. Fiebre, dolor de cabeza, escalofríos, dolor articular, tos seca.
3. Fiebre, falta de apetito, estornudos.

Recuerda que en cualquier momento puedes solicitar explicación de por qué está haciendo una pregunta en concreto.